

ネットワーク・ゲーム構築モデルへのゲーム ・コンテンツ組み込み事例

片 山 益 男

How to add a game-content to the network-game model structure: an example

KATAYAMA Masuo

目 次

1. はじめに
2. ゲームのルール
3. ネットワーク・ゲーム化への取組み方
4. 保持データ
5. ゲーム開始時の条件と初期値設定
6. 各ステップの処理内容
7. 考察
8. おわりに

Abstract

This paper presents how to add a game content to the network-game model structure (developed by the author) with an AV shop example.

キーワード：経営ゲーム，ネットワーク利用，教育用，競争型

Key Words : business game, network application, for education, competitive type

1. はじめに

本稿では、先に発表した「教育用ネットワーク・ゲームの一構築モデル」[1] に具体的なゲーム・コンテンツを組み込んだ1事例を説明する。主たる意図は、ネットワーク・ゲーム構築モデルへのコンテンツ組み込み方法の説明にある。その意味で、先の発表を補完するものである。

さらに、ネットワーク・ゲームで何を教育できるのかについて、多少の考察を加えた。

2. ゲームのルール

(1) 概要

使用したゲーム・コンテンツは、すでに何度も発表している「レコード店の経営ゲーム」である[2]。その概要を簡単に紹介すると、商品(レコード、現在ではDVDあたりを想定するのが良いかも知れない)を仕入れて、販売するという、ごく単純な活動であるが、その際に次の項目について意思決定を行うことによりゲームの優劣を評価する：1) 販売方法(現金売、掛売、手形売)と広告の有無、2) 陳列台増設(品揃え規模の制約条件)、3) 仕入枚数(在庫規模=品揃え規模)、4) 借入金増減(資金繰り)。

このゲームはもともと非競争型(スタンド・アロン型)として作成されている。すなわちゲームの途中で、他のプレイヤーなどの影響を受けることなく、ゲームを進行させる方式を採っている。しかし、同一の初期条件から出発して、各人が独自の意思決定をおこなうことにより異なる成果(利益額など)が実現されるので、その成果によってゲームの優劣を競うことはできる。

(2) 規則の変更点

今回は競争型にするため、各プレイヤーに与えられる需要量は、外部与件としての総需要を、過去の売上規模に応じて配分するという規則に変更した(非競争型の場合は、需要量を「基本需要」として、単に外部与件として設定していた)。その他のゲームの規則は変更せずに用いた。なお、競争型とは、複数のプレイヤーの振るまいが、相互に干渉し合うタイプを指す。例えば、ゼロサム・ゲームのように、あるプレイヤーの売上が増加すれば、その分だけ他のプレイヤーの売上が減少するというような状況下でのゲームである。

競争型ゲームにおいては、参加者数が複数となるが、その人数を固定するのは実際的でないので、今回提案したゲーム構成法[1]の前提通り、参加者数は任意とした。その場合、総需要をどのようにして設定するかという問題に対し、当初のゲームでは「基本需要」として設定していた外部与件の数値に、参加人数を乗じるという方法で対処した。途中で脱落者が発生した場合、当初の参加者数を基に計算した大きさを6割、実参加者数で計算した値を4割とした加重平均値を用いた。

基本総需要の配分は、各プレイヤーの売上実績により比例配分するという計算法とした。

これらの計算法は競争型ゲームとするために、とりあえず設定したものであり、教育目的をふまえて検討した結果ではない。

(3) 1サイクルのステップ数

元のゲームでは1サイクルを7ステップに分割し、逐次手計算で進行させる方法をとっていた。その中で意思決定が必要なのは4ステップであるが、ゲームの規則を理解していると、各ステップごとにサーバーとのやりとりは必要でなく、プレイヤー側だけで逐次決定していくことができる。すなわち、販売方法、陳列台増設、仕入枚数の決定結果を一度にサーバーに送っても全く支障がない。その結果、販売実績が明らかになった後、借入金増減の意思決定をしないといけないが、それは次のサイクルの最初にまわして、販売方法などの意思決定値と一緒に送ってもゲームの進行には何ら支障がない。そのような検討の結果、意思決定値をサーバーに送る入力モードと、その結果を受け取る参照モードは、サイクルに1つずつで良いことがわかり、今回提案した基本的なゲーム構成法 [1] をそのまま適用することとした。

3. ネットワーク・ゲーム化への取組み方

いま対象としているタイプの経営ゲームを、コンピュータを用いて実行しようとする場合の基本的な考え方は、

- 1) ゲーム進行と評価に必要な、記録しておくべきデータ種類を明らかにし、その初期値を設定する、
 - 2) 意思決定時に参照することが必要となる項目の、その時点でのデータ内容を表示する、
 - 3) 意思決定結果を入力するための入力様式を表示する、
 - 4) 意思決定データが入力されたら、エラー・チェックを行って入力データを記録・確認表示した後、意思決定データに基づき、ゲームの状態（現金、在庫などの有り高更新、売上記録など）を変更する、
 - 5) ゲームを継続する場合はステップ2)に戻る、
 - 6) ゲーム終了時は、記録しているデータを用いてゲームの成果を計算・表示する、
- という順序でゲームが進行するようにデータ定義を行い、プログラムを作成することである。

非競争型と競争型の違いは、ステップ4)で用いるデータが、各プレイヤーのものに限るか、参加プレイヤーのデータをすべて用いて、必要な計算を行うかどうかという点だけである。従って、複数プレイヤーのデータを使う部分以外は、すべて非競争型のゲームを作成するのと同じ考え方でコンピュータ化を進めていけばよい (例えば [3])。

競争型の場合、複数プレイヤーのデータが必要になるため、ネットワーク利用を前提とするのが普通であるが、その部分の計算をサーバーで行い、その他は端末側のパソコンで行うことも可能である。すなわち、端末とサーバーでの処理分担の振り分けには多様な方法がある。今回はすべてサーバーでの処理を前提とした方式を取っているので、どの時点で、どのような処理部分を実行させるのかを切り分けて、サーバーの処理モジュールを作成していくことが必要となる。

次に、今回取り上げた「レコード店の経営ゲーム」について、もう少し詳しい説明を加える。

4. 保持データ

ゲーム進行のためにサーバーで保持する必要のあるデータ項目を、各サイクル進行に必要なもの、決算に必要なもの(このゲームでは6ヶ月ごとに決算することになっている)、全体評価に必要なものに分類できる。その内容をリスト1に示した(項目リストを表示すべきであるが、紙幅を節約するためプログラム・リストのままとした)。

少し説明を加えると、「プレイヤー毎に変化する値」は参加人数だけの配列データとなる。「現サイクルの初期値」は、そのサイクル開始にあたって設定しておくべき項目値、「現サイクルのdecision結果」はプレイヤーから送信された値と補正值(例えば500枚しか収容能力がないとき、600枚仕入と入力されたときは500枚に修正)およびそれらを基に計算された結果、「サイクル中に変化した値」は現金や在庫など出入りにより変化した後の有り高で、初期値の値を更新してしまうと処理に支障が生じるため、別のエリアに記録しているもので、次のサイクルの初期値となるもの、「決算用初期値」は定期的な決算のため、その期間の初期値を記録するもの、または売上など、その期間の累計値を求めていくためのもの、である。

全体評価に必要な項目としては、とりあえず売上の総累計だけを記録した。利益の総額は最後の決算書での「繰越利益金+当期純利益」から求めることができる(決算書作成でゲーム終了となる時)。使用総資本の平均値などが必要なときは、記録項目を追加する必要がある。

配列データとして保持する項目について、添字0の位置は参加者番号との対応を間違えないため原則として利用していない。例外として、「先月の売上」「先々月の売上」「実販売枚数」の3項目については、添字0の位置に全参加者の総売上金額/枚数のデータを記録している。(配列データは各参加者の内容を示すもので、過去のサイクルの値を保持し

ているのではない点に留意が必要である。)

5. ゲーム開始時の条件と初期値設定

(1) ゲーム開始時の設定

ゲームを開始するときにはサーバーの保持記録を初期化し、開始時刻、入力・参照モードの長さなどを設定しなければならない。さらにゲーム・コンテンツに依存するものとして、この事例では、開始時の決算書データ、決算間隔、基本需要の大きさ・変化パターンなどを設定しなければならない。そのための画面を図1に、そのスクリプトをリスト2に示す。

(2) 初期値設定のタイミング

全プレイヤーが共通して使用するデータはゲーム開始時に設定すればよい。しかしプレイヤーごとに値が変化していく項目については、参加人数分の配列データとして記録しておかねばならないので、モード0→1への変更時に初期値設定を行う必要がある。たとえば現金など決算書の初期値は全プレイヤー共通であるが、そこに記録した値はサイクルの進行とともにプレイヤー毎に異なる値に変化していくので、配列データとしての記録が必要になる。

6. 各ステップの処理内容

先に発表した「教育用ネットワーク・ゲームの一構築モデル」[1]には、ゲーム進行のメカニズムを実現するためのプログラムがリスト1～11にわたり掲載されている(以降、原リストと呼ぶ)。その中に「ゲーム内容に依存した内容(#xx)」という部分が多数挿入されている。そこに、ゲーム・コンテンツに応じた内容プログラムを追加することにより、全体としてのゲーム進行が可能となる。

「レコード店の経営ゲーム」において、その内容がどのようなものかを次に列挙する。「#xx」は原リストの(#xx)番号に対応させている。

(1) 参加申込み受付(原リスト3:inform1a)

- #1 初期値を表示
- #2 最初の意思決定値入力様式を表示

(2) 参加受付モードから入力モードへの変更(原リスト4:mode0to1)

- #3 配列データに初期値設定

- # 4 ゲーム条件をDBに記録
- (3) 意思決定データの受付 (原リスト5 : inform1b)
 - # 5 送信されてきた意思決定値の受け取り
 - # 6 データが空かどうかのチェック
 - # 7 入力値に基づき, ゲーム内容の処理(レコード仕入れ代の支払い, 売上債権の回収, 陳列台の増減, 仕入れによる在庫枚数更新と買掛金設定, 支払い費用の計算)(原リスト6 : query1a)
 - # 8 意思決定値の受け取り結果を表示
- (4) 入力モードから参照モードへの変更 (原リスト7 : modelto2)
 - # 9 需要配分計算, 各人の実際販売枚数・売上高の計算, 費用の支払いと現金残高の更新
- (5) 実行結果の返送 (原リスト8 : query1b)
 - #10 (そのサイクルの) ゲーム実施結果の表示, 決算期なら決算処理(原リスト9 : gresulta)
 - #11 ゲーム実施結果の表示内容(原リスト10 : inform2a)
 - #12 (次の入力の参考として) 直前の値を表示
 - #13 次サイクルの入力様式の表示
- (6) 意思決定データの受付 (第2サイクル以降) (原リスト11 : inform2b)
 - #14 送信された入力データの漏れをチェック
 - #15 入力値に基づき, ゲーム内容の処理 (inform1bとの違いは, 借入金増減処理の追加)。

以上のスクリプト内容をリスト3に示す。決算のために必要となるスクリプトは独立したサブルーチンとなるので, リスト3から分離しリスト4に示す(gresultaから呼び出し)。

原リストに, これらの追加部分を挿入することにより, 完全なスクリプトとなり, それを用いてゲームを実行することができる(追加内容の一部には, 文脈を明示するため, 原リストと少し重複させているものがある)。

ゲーム実施の準備として, ユーザ登録やゲームの実施記録をとるためのデータベースを作成しておくことも必要になる。ここでは使用経験のあるMS-Accessを用いた。サーバーに, webアプリケーション用の仮想ディレクトリを作成し, 以上のスクリプト・ファイルとDBファイルをそのフォルダに置く。

実際にゲームを実施するには、まずgamesetting.asp（リスト2）にアクセスして、ゲームを開始状態にする。そして各参加者はgjoin.asp（参加申込み [1]）を呼び出せば、ゲームに参加可能となる。

ゲームが開始されると、各プレイヤーは、次のモード変更の時刻まであとどれだけの時間があるのかを知りたくなる（一応モード変更時刻は表示されているが、いちいち計算せずに、残り時間を知りたいという欲求が発生する）。そのようなとき、リスト5に示したスクリプトを起動すれば、簡単に残り時間を知ることができる。

教室などに集合してゲームを実施している場合、全員が現モードの作業を終了し、早く次のモードに切り替えたいという状況が生じることも考えられる。その場合はリスト6のスクリプトを起動すれば、すぐにモード変更が可能となる。ただし、その利用は集合利用において、そのリーダーのみに限る必要がある。

7. 考 察

競争型のゲームは、非競争型のゲームと比べて、教育効果という立場から、どのような特徴があるのか、その問いに答えることが必要なのであるが、今回紹介したゲーム・コンテンツについて、その理由を明確に説明できるまでに至っていない。今回取り上げたものは、ネットワーク・ゲームの進行メカニズムがうまく働くかどうかの検証を第一義としたためである。しかし、競争型のゲームとして利用するという採択過程において、多少の検討をしたので、その内容を説明する。

非競争型であっても、そのゲームにより、どのような教育効果をあげることができるのかに真正面から取り組むと、その説明は簡単ではない。今回対象とした「レコード店経営」ゲームの開発当初の目的は、（日々の業務が利益実現にどのように関係しているかという）経営の基本的なメカニズムを理解させるということであった。その限りにおいて、このゲームは多くの学生からの支持を得て、成功したものと思っている。

しかし、このゲームを用いて、どうすればより高い利益をあげることができるのか、その規則を探索し、発見された規則が経営一般に適用できるものだと誤解されると、教育効果はなくなってしまふ。利益追求のための規則はゲームに組み込んだ規則に依存し、それを経営一般に拡張することはできないからである。例えばこのゲームでは借金を増やしても規模拡大のほうが有利になるような規則となっているが、借り入れコストを上げるだけで、そのような規則は適用できなくなる。

同じような意味で、競争型ゲームで何を教育することができるのだろうか。そのゲーム

に組み込まれた競争の規則を理解し、ゲームで有利に振る舞うことができたとしても、それを競争的環境にある経営一般に拡張することはできない。逆に、例えば「ゲーム理論」を学習させるために競争型ゲームを用いるとすれば、具体的にどのようなゲームを構築すればよいのか筆者には勉強不足のため不明である。少なくともプレイヤー間で協調や対立が可能なようなゲーム環境を準備しなければならないであろう。

このような理由で、教育目的にかなう競争型ゲームをどのように構築すればよいのか、筆者にはまだ解答は見つかっていない。一つ言えそうなことは、競争型ゲームは非競争ゲームより、学習者の取り組み意欲を高めることができそうだということである。かつて「レコード店経営」ゲームの実行にコンピュータを利用する方法を考案し、その発表の中で「この方法ではプレイヤーのゲームへの取り組みが表面的になる可能性がある」と指摘した[3]。その場合と比べると、競争型では、相手が存在するため、真剣に取り組もうという意欲が増加するよう感じられる（定量的に比較調査したものではないが）。

8. おわりに

提案したネットワークによるゲームが実際に利用できることを示すために、一つのゲーム・コンテンツの組み込み例を紹介した。多少の学習で、誰でもゲーム構築が可能になるので、試行していただければ幸いである。

参考文献

- 拙著 [2000], 『経営システムと情報システム』, 中央経済社, 147-165頁。
拙稿 [2004], 「webページを用いた教育用ゲーム開発事例」, 大阪産業大学経営論集第5巻第3号, 17-40頁。
拙稿 [2006], 「教育用ネットワーク・ゲームの一構築モデル」, 大阪産業大学経営論集第7巻第3号, 39-70頁。

図1 ゲーム開始条件の設定ページ画面

新しいゲームの開始設定

現モード番号：2

現モードはゲーム開始設定には不適當です。

(前のゲームが進行途中の可能性がります。)

開始設定をしますか？ y/n： [y]

ゲームのサイクル数（終了までの回数）： [12]

入力モードの長さ（分）： [2]

参照モードの長さ（分）： [2]

現在（開始設定）から参加メ切までの時間（分）： [3]

現在の時刻は 2006/02/25 10:23:23

期首貸借対照表の数値など

現金 [2200] 買掛金 [1200]

売掛金 [1700] 借入金 [6000]

受取手形 [0] 納税引当金 [280]

商品 [2700] 資本金 [2000]

什器備品 [1600] 繰越利益金 [240]

構築物 [1800] 当期純利益 [280]

1期の減価償却額 [300]

固定費用／サイクル [530]

1決算あたりのサイクル数 [6]

需要規模と広告効果の大きさ

基本需要の初期値 [700]

広告効果の大きさ (%) [15]

需要の推移：

横ばい 上昇 下降

変化の型（差／比）と、サイクルあたりの変化数：

変化差（枚） 変化率（%）

変化枚数／% [10]

[設定]

リスト1 Application変数の一覧リスト表示 apvarlst.asp

```

<HTML>
<HEAD>
<META HTTP-EQUIV="Content-Type" CONTENT="text/html; charset=shift_jis">
<TITLE>Application variable list</TITLE>
<STYLE TYPE="text/css">
</STYLE>
</HEAD>
<body bgcolor="#ffffff">
ゲームID Application("gameid")= <%= Application("gameid") %><br>
サイクル数 Application("cyclelmt")= <%= Application("cyclelmt") %><br>
入力期間幅 Application("inspan")= <%= Application("inspan") %> 分<br>
参照期間幅 Application("seespan")= <%= Application("seespan") %> 分<br><br>

サイクル番号Application("cycleno")= <%= Application("cycleno") %><br>
モード番号 Application("pmode")= <%= Application("pmode") %><br>
締切時刻 Application("timelmt")= <%= Application("timelmt") %><br>
次の締切時刻Application("nextlmt")= <%= Application("nextlmt") %><br><br>

参加者数 Application("joinsize")= <%= Application("joinsize") %><br>
継続者数 Application("donesize")= <%= Application("donesize") %><br>
参加状況 Application("s1")= <%= Application("s1") %> 2 以上は脱落<br>
参加番号 Application("joinno")= <%= Application("joinno") %><br>
エラーマーク Application("errsw")= <%= Application("errsw") %><br><br>
<HR>
** ゲーム内容依存項目<br><br>
* 全プレイヤー共通値<br>
基本需要枚数Application("bdemand")= <%= Application("bdemand") %><br>
広告効果 Application("adef")= <%= Application("adef") %> %<br><br>
BS初期値1 Application("init1")= <%= Application("init1") %><br>
BS初期値2 Application("init2")= <%= Application("init2") %><br>
固定支払費用Application("fixexpns")= <%= Application("fixexpns") %><br>
1期償却額 Application("depress")= <%= Application("depress") %><br>
償却実施sw Application("asetsw")= <%= Application("asetsw") %><br>
決算間隔 Application("bsspan")= <%= Application("bsspan") %><br><br>
* プレイヤー毎に変化する値 (配列データ) <br><br>
需要配分計算用<br>
先々月の売上Application("sv0")= <%= Application("sv0") %> 枚<br>
先月の売上 Application("sv1")= <%= Application("sv1") %><br><br>
現サイクルの初期値<br>

```

現金 Application("j1") = <%= Application("j1") %> 千円

 売掛金 Application("j2") = <%= Application("j2") %>

 受取手形 1 Application("j3") = <%= Application("j3") %>

 受取手形 2 Application("j4") = <%= Application("j4") %> 次月入金

 在庫数量 Application("j5") = <%= Application("j5") %>

 買掛金 Application("j6") = <%= Application("j6") %>

 借入金 Application("j7") = <%= Application("j7") %>

 +台数 1 Application("j11") = <%= Application("j11") %> 今月

 +台数 2 Application("j12") = <%= Application("j12") %> 次月

 +台数 3 Application("j13") = <%= Application("j13") %>

 +台数 4 Application("j14") = <%= Application("j14") %>

 +台数 5 Application("j15") = <%= Application("j15") %>

 現サイクルのdecision結果

 販売方法 Application("d1") = <%= Application("d1") %> 0.9:現金 1.1:手形

 広告有無 Application("d2") = <%= Application("d2") %> 1:有り

 陳列台台数 Application("d3") = <%= Application("d3") %>

 仕入数量 Application("d4") = <%= Application("d4") %>

 借入増減額 Application("d5") = <%= Application("d5") %>

 実仕入数量 Application("r1") = <%= Application("r1") %>

 基本需要数 Application("r2") = <%= Application("r2") %>

 実販売枚数 Application("r3") = <%= Application("r3") %>

 実陳列台数 Application("r4") = <%= Application("r4") %>

 陳列台費用 Application("r5") = <%= Application("r5") %>

 広告費 Application("r6") = <%= Application("r6") %>

 支払利息 Application("r7") = <%= Application("r7") %>

 サイクル中に変化した有り高 (次サイクルの初期値となる)

 次現金 Application("n1") = <%= Application("n1") %>

 次売掛金 Application("n2") = <%= Application("n2") %>

 次受取手形 1 Application("n3") = <%= Application("n3") %>

 次受取手形 2 Application("n4") = <%= Application("n4") %>

 次在庫数量 Application("n5") = <%= Application("n5") %>

 次買掛金 Application("n6") = <%= Application("n6") %>

 次借入金 Application("n7") = <%= Application("n7") %>

 次+台数 1 Application("n11") = <%= Application("n11") %>

 次+台数 2 Application("n12") = <%= Application("n12") %>

 次+台数 3 Application("n13") = <%= Application("n13") %>

 次+台数 4 Application("n14") = <%= Application("n14") %>

 次+台数 5 Application("n15") = <%= Application("n15") %>

 決算用初期値

 商品在庫額 Application("bs1") = <%= Application("bs1") %>

 固定資産有高Application("bs2") = <%= Application("bs2") %> (全プレイヤー共通)


```

納税引当金 Application("bs3") = <%= Application("bs3") %><br>
資本金      Application("bs4") = <%= Application("bs4") %> (全プレイヤー共通) <br>
繰越利益   Application("bs5") = <%= Application("bs5") %><br>
純利益税引 Application("bs6") = <%= Application("bs6") %><br>
売上累計   Application("cm1") = <%= Application("cm1") %><br>
仕入累計   Application("cm2") = <%= Application("cm2") %><br>
費用累計   Application("cm3") = <%= Application("cm3") %><br>
売上総累計 Application("tsales") = <%= Application("tsales") %><br>
</body>
</html>

```

リスト2 ゲーム開始条件の設定 gamesetting.asp

```

<HTML>
<HEAD>
<META HTTP-EQUIV="Content-Type" CONTENT="text/html; charset=shift_jis">
<TITLE>ゲーム開始の設定</TITLE>
<STYLE TYPE="text/css">
</STYLE>
</HEAD>
<BODY><br>
新しいゲームの開始設定<br><br>
現モード番号 :
<%= Application("pmode") %><br><br>
<% ' 開始設定をしても良いかの確認
If Application("pmode") < 90 Then
    Response.Write("現モードはゲーム開始設定には不適當です。<br><br>")
    Response.Write(" (前のゲームが進行途中の可能性がります。)")
End If
If Request.Form("yn") ="y" Then
    ' ゲーム開始の設定
    Application("pmode") = 0
    Application("joinsize") = 0
    Application("donesize") = 0
    Application("cycleno") = 0
    clmt = CInt(Request.Form("ncycle1"))
    Application("cyclelmt") = clmt
    Application("inspan") = CInt(Request.Form("inspan1"))
    Application("seespan") = CInt(Request.Form("seespan1"))
    min2 = Request.Form("min1")

```

```

h1 = Hour(Now)
m1 = Minute(Now)
now1 = DateValue(Now) + TimeSerial(h1,m1,0)
Application("timelmt") = DateAdd("n", min2, now1)
Application("nextlmt") = DateAdd("n", Application("inspan"), Application("timelmt"))
yy = Year(Application("timelmt"))
mm = Month(Application("timelmt"))
dd = Day(Application("timelmt"))
ymd = Dateserial(yy, mm, dd)
Application("gameid") = "g" & ymd & "_" & Hour(Application("timelmt"))
inn1 = Request.Form("in1")
inn2 = Request.Form("in2")
inn3 = Request.Form("in3")
inn4 = Request.Form("in4")
inn5 = Request.Form("in5")
inn6 = Request.Form("in6")
inn11 = Request.Form("in11")
inn12 = Request.Form("in12")
inn13 = Request.Form("in13")
inn14 = Request.Form("in14")
inn15 = Request.Form("in15")
inn16 = Request.Form("in16")
inn21 = Request.Form("in21")
inn22 = Request.Form("in22")
inn23 = Request.Form("in23")
trend1 = Request.Form("trend")
pat1 = Request.Form("pat")
Application("bsspan") = CInt(Request.Form("bsspan"))
Application("depress") = CInt(Request.Form("depress"))
Application("fixexpns") = CInt(Request.Form("fixexpns"))
Application("bs2") = 0 + inn5 + inn6
Application("bs4") = CInt(inn14)
ary1 = Array(" ",inn1,inn2,inn3,inn4,inn5,inn6)
ary2 = Array(" ",inn11,inn12,inn13,inn14,inn15,inn16)
Application("init1") = join(ary1,",")
Application("init2") = join(ary2,",")
' 広告効果の大きさを記録
Application("adef") = inn22
' 各サイクルの基本需要枚数を設定
dlist = ""
If trend1 = "1" Then

```

```
For i = 1 to clmt
    dlist = dlist & "," & inn21
Next
Elseif trend1 = "2" Then
    val = inn21 + 0
    If pat1 = "1" Then
        sa = inn23 + 0
        For i = 1 to clmt
            sval = CStr(val)
            dlist = dlist & "," & sval
            val = val + sa
        Next
    Else
        For i = 1 to clmt
            hi = inn23 + 0
            sval = CStr(int(val))
            dlist = dlist & "," & sval
            val = val * (1 + hi/100)
        Next
    End If
Elseif trend1 = "3" Then
    val = inn21 + 0
    If pat1 = "1" Then
        sa = inn23 + 0
        For i = 1 to clmt
            sval = CStr(val)
            dlist = dlist & "," & sval
            val = val - sa
        Next
    Else
        For i = 1 to clmt
            hi = inn23 + 0
            sval = CStr(int(val))
            dlist = dlist & "," & sval
            val = val * (1 - hi/100)
        Next
    End If
Else
    Response.Write("需要トレンド設定エラー")
End If
Application("bdemand") = dlist
```

```
' 登録者の参加状態をクリア
sql = "update user set joinno = " & 0 & ", drop = " & "0" & " "
Set adoConn = Server.CreateObject("ADODB.Connection")
Set adoRecs= Server.CreateObject("ADODB.Recordset")
adoConn.Open "netgame"
adoConn.Execute sql
adoConn.Close
End If
' その他のApplication変数を初期化
Application("errsw") = 0
Application("j1") = " "
Application("j2") = " "
Application("j3") = " "
Application("j4") = " "
Application("j5") = " "
Application("j6") = " "
Application("j7") = " "
Application("j11") = " "
Application("j12") = " "
Application("j13") = " "
Application("j14") = " "
Application("j15") = " "
Application("d1") = " "
Application("d2") = " "
Application("d3") = " "
Application("d4") = " "
Application("d5") = " "
Application("r1") = " "
Application("r2") = " "
Application("r3") = " "
Application("r4") = " "
Application("r5") = " "
Application("r6") = " "
Application("r7") = " "
Application("s1") = " "
Application("n1") = " "
Application("n2") = " "
Application("n3") = " "
Application("n4") = " "
Application("n5") = " "
Application("n6") = " "
```

```
Application("n7")=" "  
Application("n11")=" "  
Application("n12")=" "  
Application("n13")=" "  
Application("n14")=" "  
Application("n15")=" "  
Application("sv0")=" "  
Application("sv1")=" "  
Application("bs1")=" "  
Application("bs3")=" "  
Application("bs5")=" "  
Application("bs6")=" "  
Application("cm1")=" "  
Application("cm2")=" "  
Application("cm3")=" "  
Application("tsales")=" "  
Application("asetsw")= 0  
Session.Abandon  
%>  
<HR>  
<FORM NAME="fm" METHOD="post" ACTION="gamesetting.asp">  
<TABLE BORDER="0">  
<TR>  
<TD>開始設定をしますか? y/n: </TD>  
<TD><INPUT NAME="yn" TYPE="text" VALUE="y"></TD>  
</TR>  
<TR>  
<TD>ゲームのサイクル数 (終了までの回数): </TD>  
<TD><INPUT NAME="ncycle1" TYPE="text" VALUE="12"></TD>  
</TR>  
<TR>  
<TD>入力モードの長さ (分): </TD>  
<TD><INPUT NAME="inspan1" TYPE="text" VALUE="2"></TD>  
</TR>  
<TR>  
<TD>参照モードの長さ (分): </TD>  
<TD><INPUT NAME="seespan1" TYPE="text" VALUE="2"></TD>  
</TR>  
<br><br>  
<TR>  
<TD>現在 (開始設定) から参加切までの時間 (分): </TD>
```



```

<TD><INPUT NAME="min1" TYPE="text" VALUE="3"></TD>
</TR>
</TABLE><br>
現在の時刻は
<%= Now %><br><br>
<HR>
期首貸借対照表の数値など
<br><br>
<TABLE BORDER="0">
<TR>
<TD>現金</TD>
<TD><INPUT NAME="in1" TYPE="text" VALUE="2200"></TD>
<TD></TD>
<TD>買掛金</TD>
<TD><INPUT NAME="in11" TYPE="text" VALUE="1200"></TD>
</TR>
<TR>
<TD>売掛金</TD>
<TD><INPUT NAME="in2" TYPE="text" VALUE="1700"></TD>
<TD></TD>
<TD>借入金</TD>
<TD><INPUT NAME="in12" TYPE="text" VALUE="6000"></TD>
</TR>
<TR>
<TD>受取手形</TD>
<TD><INPUT NAME="in3" TYPE="text" VALUE="0"></TD>
<TD></TD>
<TD>納税引当金</TD>
<TD><INPUT NAME="in13" TYPE="text" VALUE="280"></TD>
</TR>
<TR>
<TD>商品</TD>
<TD><INPUT NAME="in4" TYPE="text" VALUE="2700"></TD>
<TD></TD>
<TD>資本金</TD>
<TD><INPUT NAME="in14" TYPE="text" VALUE="2000"></TD>
</TR>
<TR>
<TD>什器備品</TD>
<TD><INPUT NAME="in5" TYPE="text" VALUE="1600"></TD>
<TD></TD>

```

```

<TD>繰越利益金</TD>
<TD><INPUT NAME="in15" TYPE="text" VALUE="240"></TD>
</TR>
<TR>
<TD>構築物</TD>
<TD><INPUT NAME="in6" TYPE="text" VALUE="1800"></TD>
<TD></TD>
<TD>当期純利益</TD>
<TD><INPUT NAME="in16" TYPE="text" VALUE="280"></TD>
</TR>
</TABLE><br>
1期の減価償却額 <INPUT NAME="depress" TYPE="text" VALUE="300"><br>
固定費用／サイクル <INPUT NAME="fixexpsn" TYPE="text" VALUE="530"><br><br>
1決算あたりのサイクル数 <INPUT NAME="bsspan" TYPE="text" VALUE="6"><br><br>
<HR>
需要規模と広告効果の大きさ<br><br>
    基本需要の初期値 <INPUT NAME="in21" TYPE="text" VALUE="700"><br><br>
    広告効果の大きさ (%) <INPUT NAME="in22" TYPE="text" VALUE="15"><br><br>
需要の推移 : <br><br>
    横ばい <INPUT TYPE = "radio" NAME="trend" VALUE = "1" Checked>
    上昇 <INPUT TYPE = "radio" NAME="trend" VALUE = "2">
    下降 <INPUT TYPE = "radio" NAME="trend" VALUE = "3"><br><br>
変化の型 (差／比) と, サイクルあたりの変化数 : <br><br>
    変化差 (枚) <INPUT TYPE = "radio" NAME="pat" VALUE = "1" Checked>
    変化率 (%) <INPUT TYPE = "radio" NAME="pat" VALUE = "2"><br><br>
    変化枚数／% <INPUT NAME="in23" TYPE="text" VALUE="10"><br><br>
<INPUT TYPE="submit" VALUE="設定">
</FORM>
<br><br>
</BODY>
</HTML>

```

リスト3 ゲーム・コンテンツ「レコード店の経営」の組み込みにより追加した部分
(番号は原リストの追加番号に対応)

```

<リスト3 inform1a>
# 1
<%
in1 = split(Application("init1"),",")

```

```

in2 = split(Application("init2"),",")
%>
<HR>
ゲームの初期値は次のようになっています。<br><br>
<TABLE>
<TR>
<TD>現金</TD>
<TD><%= in1(1) %></TD>
<TD> | </TD>
<TD>買掛金</TD>
<TD><%= in2(1) %></TD>
</TR>
<TR>
<TD>売掛金</TD>
<TD><%= in1(2) %></TD>
<TD> </TD>
<TD>借入金</TD>
<TD><%= in2(2) %></TD>
</TR>
<TR>
<TD>受取手形</TD>
<TD><%= in1(3) %></TD>
<TD></TD>
<TD>納税引当金</TD>
<TD><%= in2(3) %></TD>
</TR>
<TR>
<TD>商品</TD>
<TD><%= in1(4) %></TD>
<TD></TD>
<TD>資本金</TD>
<TD><%= in2(4) %></TD>
</TR>
<TR>
<TD>什器備品</TD>
<TD><%= in1(5) %></TD>
<TD></TD>
<TD>繰越利益金</TD>
<TD><%= in2(5) %></TD>
</TR>
<TR>

```

```

<TD>構築物</TD>
<TD><%= in1(6) %></TD>
<TD></TD>
<TD>当期純利益</TD>
<TD><%= in2(6) %></TD>
</TR>
<TR>
<TD>在庫数量</TD>
<TD><% = int(in1(4)/1.5) %></TD>
<TD></TD>
</TR>
</TABLE><br><br>

```

2

```

<TABLE BORDER="0">
<TR>
<TD>販売方法：</TD>
</TR>
<TR>
<TD> 現金売</TD>
<TD><INPUT TYPE = "radio" NAME="sale" VALUE = "0.9"></TD>
<TD> 掛売</TD>
<TD><INPUT TYPE = "radio" NAME="sale" VALUE = "1"></TD>
<TD> 手形売</TD>
<TD><INPUT TYPE = "radio" NAME="sale" VALUE = "1.1"></TD>
</TR>
<TD>広告の有無：</TD>
</TR>
<TR>
<TD> 出す</TD>
<TD><INPUT TYPE = "radio" NAME="ad" VALUE = "1"></TD>
<TD> 出さない</TD>
<TD><INPUT TYPE = "radio" NAME="ad" VALUE = "0"></TD>
</TR>
</TABLE><br><br>
<TABLE BORDER="0">
<TR>
<TD>陳列台台数：</TD>
<TD><INPUT NAME="dai" TYPE="text" VALUE="4"></TD>
</TR>
<TR>

```

```
<TD>仕入枚数:</TD>
<TD><INPUT NAME="invol" TYPE="text" ></TD>
</TR>
</TABLE><br>
```

<リスト4 mode0to1>

```
# 3
inn = split(Application("bdemand"),",")
svoll = inn(1)
inn = split(Application("init1"),",")
'商品金額を在庫数量に変換
zaikosu = Cstr(int(inn(4)/1.5))
'初期値設定
st = ""
zr = ""
in1 = ""
in2 = ""
in3 = ""
in4 = ""
n = Application("joinsize")
in5 = CStr(svoll * n)
in9 = ""
for i = 1 to n
    st = st & ","
    zr = zr & "," & "0"
    in1 = in1 & "," & inn(1)
    in2 = in2 & "," & inn(2)
    in3 = in3 & "," & inn(3)
    in4 = in4 & "," & inn(4)
    in5 = in5 & "," & svoll
    in9 = in9 & "," & zaikosu
next
Application("null") = st
Application("zero") = zr
Application("j1") = in1
Application("j2") = in2
Application("j3") = in3
Application("j4") = zr
Application("j5") = in9
Application("bs1") = in4
Application("sv0") = in5
```

```
Application("sv1") = in5
inn = split(Application("init2"), ",")
in1 = " "
in2 = " "
in3 = " "
in5 = " "
in6 = " "
for i = 1 to n
  in1 = in1 & "," & inn(1)
  in2 = in2 & "," & inn(2)
  in3 = in3 & "," & inn(3)
  in5 = in5 & "," & inn(5)
  in6 = in6 & "," & inn(6)
next
Application("bs3") = in3
Application("bs5") = in5
Application("bs6") = in6
Application("j6") = in1
Application("j7") = in2
Application("j11") = zr
Application("j12") = zr
Application("j13") = zr
Application("j14") = zr
Application("j15") = zr
Application("d1") = zr
Application("d2") = zr
Application("d3") = zr
Application("d4") = zr
Application("d5") = zr
Application("r1") = zr
Application("r2") = zr
Application("r3") = zr
Application("r4") = zr
Application("r5") = zr
Application("r6") = zr
Application("r7") = zr
Application("s1") = zr
Application("n1") = zr
Application("n2") = zr
Application("n3") = zr
Application("n4") = zr
```

```
Application("n5") = zr
Application("n6") = zr
Application("n7") = Application("j7")
Application("n11") = zr
Application("n12") = zr
Application("n13") = zr
Application("n14") = zr
Application("n15") = zr
Application("cm1") = zr
Application("cm2") = zr
Application("cm3") = zr
Application("tsales") = zr
Application UnLock
%>
```

4

<% 'ゲーム条件をDBに記録

```
Set adoConn = Server.CreateObject("ADODB.Connection")
Set adoRecs= Server.CreateObject("ADODB.Recordset")
adoConn.Open "netgame"
adoRecs.Open "gamelog", adoConn, 3, 3, 2
adoRecs.AddNew
adoRecs("gameid").Value = Application("gameid")
adoRecs("cyclelmt").Value = Application("cyclelmt")
adoRecs("joinsize").Value = Application("joinsize")
adoRecs("bsspan").Value = Application("bsspan")
adoRecs("init1").Value = Application("init1")
adoRecs("init2").Value = Application("init2")
adoRecs("bdemand").Value = Application("bdemand")
adoRecs("adef").Value = Application("adef")
adoRecs("fixexpns").Value = Application("fixexpns")
adoRecs("depress").Value = application("depress")
adoRecs.Update
adoRecs.Close
adoConn.Close
```

%>

<リスト5 inform1b>

5

```
sale1 = Request.Form("sale")
ad1 = Request.Form("ad")
```

```
dai1 = Request.Form("dai")
invol1 = Request.Form("invol")
```

6

```
' すべての項目にデータが入力されているか?
If sale1 <> "" and ad1 <> "" and dai1 <> "" and invol1 <> "" Then
Else
  Application("errsw") = 1
  Response.Write("入力漏れの項目があります。追加後, 再送信してください<br><br>")
End If
```

7

```
' 現金支払いと入金計算
in1 = split(Application("j1"),",")
in2 = split(Application("j2"),",")
in3 = split(Application("j3"),",")
in4 = split(Application("j4"),",")
in5 = split(Application("j5"),",")
in6 = split(Application("j6"),",")
in7 = split(Application("j7"),",")
in21 = split(Application("n1"),",")
in22 = split(Application("n2"),",")
in23 = split(Application("n3"),",")
in24 = split(Application("n4"),",")
in25 = split(Application("n5"),",")
in26 = split(Application("n6"),",")
in27 = split(Application("n7"),",")
' 仕入代金の支払い
sa = in1(jno1) - in6(jno1)
if sa < 0 Then
  Response.Write("仕入代金の支払いができません。倒産です。<br><br>")
  Response.Write("このゲームを続行できません。<br><br>")
  Application("errsw") = 1
  wtr1(jno1) = "9"
Else
  in21(jno1) = sa
  in26(jno1) = 0
End If
End If
If Application("errsw") = 0 Then
' 売上債権の回収
```



```

in21(jno1) = in21(jno1) + in2(jno1) + in3(jno1)
in23(jno1) = in4(jno1)
in22(jno1) = 0
in24(jno1) = 0
Application("n1") = join(in21,",")
Application("n2") = join(in22,",")
Application("n3") = join(in23,",")
Application("n4") = join(in24,",")
' 販売方法と広告の有無を記録
in1 = split(Application("d1"),",")
in2 = split(Application("d2"),",")
in3 = split(Application("d3"),",")
in4 = split(Application("d4"),",")
in1(jno1) = sale1
in2(jno1) = ad1
in3(jno1) = dai1
in4(jno1) = invol1
Application("d1") = join(in1,",")
Application("d2") = join(in2,",")
Application("d3") = join(in3,",")
Application("d4") = join(in4,",")
' 陳列台台数の修正
dai1 = dai1 + 0
If dai1 < 4 Then
    Response.Write("陳列台台数は4台を下回ることはできません。。 <br><br>")
    dai1 = 4
End If
' 最大許容増加台数を設定
maxdai = 4
sa = maxdai + 4 - dai1
' 指定台数は最大許容増加台数を超過しているか？
' 指定台数と最大許容台数の小さい方（実際台数）を dai2 に記録
If sa < 0 Then
    Response.Write("陳列台台数を希望台数まで増加させることはできません。。 <br><br>")
    dai2 = maxdai + 4
Else
    dai2 = dai1 + 0
End If
' 先月から持ち越しの追加台数がある場合、今月の指定台数 dai2 は持ち越し台数を許容できるか？
in11 = split(Application("j11"),",")

```

```

sa = dai2 - 4 - in11(jno1)
If sa < 0 Then
  Response.Write("陳列台台数を減少させることはできません。。 <br><br>")
  dai2 = 4 + in11(jno1)
  sa = 0
End If
' sa > 0 : 先月より台数が増加となる場合, または先月からの持ち越し台数がある場合,
' 将来の持ち越し必要台数を記録
If sa > 0 or in11(jno1) > 0 Then
  in12 = split(Application("j12"),",")
  in13 = split(Application("j13"),",")
  in14 = split(Application("j14"),",")
  in15 = split(Application("j15"),",")
  in31 = split(Application("n11"),",")
  in32 = split(Application("n12"),",")
  in33 = split(Application("n13"),",")
  in34 = split(Application("n14"),",")
  in35 = split(Application("n15"),",")
  in31(jno1) = sa + in12(jno1)
  in32(jno1) = sa + in13(jno1)
  in33(jno1) = sa + in14(jno1)
  in34(jno1) = sa + in15(jno1)
  in35(jno1) = sa
  Application("n11") = join(in31,",")
  Application("n12") = join(in32,",")
  Application("n13") = join(in33,",")
  Application("n14") = join(in34,",")
  Application("n15") = join(in35,",")
End If
' 在庫可能枚数, 仕入可能枚数の計算
maxinvol = dai2 * 600 - in5(jno1)
involl = involl+0
If involl > maxinvol Then
  Response.Write("希望枚数を仕入れることはできません。。 <br><br>")
  invol2 = maxinvol
Else
  invol2 = involl
End If
' 買掛金の計算
in26(jno1) = int(involl * 1.5)
Application("n6") = join(in26,",")

```

```

' 在庫数量の更新
in25(jno1) = in5(jno1) + invol2
Application("n5") = join(in25,")
' 実際仕入数量, 陳列台台数, 陳列台費用, 広告費, 支払利息の記録
in1 = split(Application("r1"),",")
in2 = split(Application("r4"),",")
in3 = split(Application("r5"),",")
in4 = split(Application("r6"),",")
in5 = split(Application("r7"),",")
in1(jno1) = invol2
in2(jno1) = dai2
in3(jno1) = (dai2 - 4) * 10
in4(jno1) = 100 * ad1
in5(jno1) = int(in7(jno1) * 0.01)
Application("r1") = join(in1,")
Application("r4") = join(in2,")
Application("r5") = join(in3,")
Application("r6") = join(in4,")
Application("r7") = join(in5,")

```

<リスト6 query1a>

```

# 8
t1 = split(Application("d1"),",")
t2 = split(Application("d2"),",")
t3 = split(Application("d3"),",")
t4 = split(Application("d4"),",")
t5 = split(Application("r4"),",")
t6 = split(Application("r1"),",")
t7 = split(Application("n5"),",")
t8 = split(Application("d5"),",")
k = CStr(t8(j))
%>
<HR>
入力された意思決定データは次のように処理されました。<br><br>
<%
If Application("cycleno") > 1 Then
  Response.Write("前サイクルの借入金増減額： " & k & "<br><br>")
End If
%>
<TABLE>
<TR>

```

```

<TD> </TD>
<TD>入力値</TD>
<TD> | </TD>
<TD>修正値</TD>
</TR>
<TR>
<TD>販売方法 </TD>
<TD><%= t1(j) %></TD>
<TD> </TD>
</TR>
<TR>
<TD>広告 </TD>
<TD><%= t2(j) %></TD>
<TD> </TD>
<TD> </TD>
</TR>
<TR>
<TD>陳列台数 </TD>
<TD><%= t3(j) %></TD>
<TD> </TD>
<TD><%= t5(j) %></TD>
</TR>
<TR>
<TD>仕入数量 </TD>
<TD><%= t4(j) %></TD>
<TD> </TD>
<TD><%= t6(j) %></TD>
</TR>
<TR>
<TD>在庫数量 </TD>
<TD><%= t7(j) %></TD>
<TD> </TD>
</TR>
</TABLE><br><br>

```

<リスト7 modelto2>

9

<%

' 需要配分計算

in1 = split(Application("sv0"),",")

in2 = split(Application("sv1"),",")

```

in3 = split(Application("bdemand"),",")
wtr1 = split(Application("s1"),",")
n = Application("joinsize")
d = Application("dropsize")
j = Application("cycleno")
btdmd = in3(j) * (n * 0.6 + (n - d) * 0.4)
btdmd = int(btdmd + 0.9)
rtdmd = btdmd * 0.6 + in2(0) * 0.4
rtdmd = int(rtdmd + 0.9)
ratioid = rtdmd / (in1(0) * 0.4 + in2(0) * 0.6)
in4 = split(Application("r2"),",")
' 各人の基本需要を記録
For i = 1 to n
  If wtr1(i) = 1 Then
    in4(i) = int((in1(i) * 0.4 + in2(i) * 0.6) * ratioid)
  End If
Next
Application("r2") = join(in4,",")
' 各プレイヤーの実績計算
t1 = split(Application("d1"),",")
t2 = split(Application("d2"),",")
t3 = split(Application("r2"),",")
t4 = split(Application("r3"),",")
t5 = split(Application("n1"),",")
t6 = split(Application("n2"),",")
t7 = split(Application("n4"),",")
t8 = split(Application("n5"),",")
t9 = split(Application("r4"),",")
t10 = split(Application("r5"),",")
t11 = split(Application("r6"),",")
t12 = split(Application("r7"),",")
tsales = 0
for i = 1 to n
  If wtr1(i) = 1 Then
    hows = t1(i) + 0
    svoll1 = t3(i) * hows * (1 + Application("adeff") * t2(i) /100)
    svoll1 = int(svoll1)
    svoll2 = int(t8(i) * 0.3)
    If svoll1 < svoll2 Then
      t4(i) = svoll1
    Else

```

```

    t4(i) = svol2
  End If
  tsales = tsales + t4(i)
  t8(i) = t8(i) - t4(i)
  sales = int(t4(i) * 2.5)
  If hows = 0.9 Then
    t5(i) = t5(i) + sales
  ElseIf hows = 1 Then
    t6(i) = t6(i) + sales
  Else
    t7(i) = t7(i) + sales
  End If
  ' 費用の支払い
  t5(i) = t5(i) - (530 + t10(i) + t11(i) + t12(i))
  If t5(i) < 0 Then
    wtr1(i) = 8
  End If
End If
next
t4(0) = tsales
Application("d1") = join(t1,",")
Application("d2") = join(t2,",")
Application("r2") = join(t3,",")
Application("r3") = join(t4,",")
Application("n1") = join(t5,",")
Application("n2") = join(t6,",")
Application("n4") = join(t7,",")
Application("n5") = join(t8,",")
%>

<リスト8 query1b>
# 10
tk = int(application("cycleno")/application("bsspan")) * application("bsspan")
tk = application("cycleno") - tk
If tk = 0 Then
  Server.Execute("kessan.asp")
End If

<リスト9 gresulta>
# 11
<%

```

```

j = Application("joinno")
wtr1 = split(Application("s1"),",")
t1 = split(Application("r2"),",")
t2 = split(Application("r3"),",")
t3 = split(Application("r5"),",")
t4 = split(Application("r6"),",")
t5 = split(Application("r7"),",")
t6 = split(Application("r1"),",")
t7 = split(Application("n5"),",")
sales = int(t2(j) * 2.5)
siire = int(t6(j) * 1.5)
tcost = 0 + Application("fixexpns") + t3(j) + t4(j) + t5(j)
cm1 = split(Application("cm1"),",")
cm2 = split(Application("cm2"),",")
cm3 = split(Application("cm3"),",")
cm9 = split(Application("tsales"),",")
cm1(j) = cm1(j) + sales
cm2(j) = cm2(j) + siire
cm3(j) = cm3(j) + tcost
cm9(j) = cm9(j) + sales
Application("cm1") = join(cm1,",")
Application("cm2") = join(cm2,",")
Application("cm3") = join(cm3,",")
Application("tsales") = join(cm9,",")
If wtr1(j) = "8" Then
    Response.Write("費用支払いのための現金不足で倒産しました！<br><br>")
    wtr1(j) = "9"
    Application("s1") = join(wtr1,",")
End If
%>
<TABLE>
<TR>
<TD>基本需要枚数</TD>
<TD><%= t1(j) %></TD>
</TR>
<TR>
<TD>実際販売枚数</TD>
<TD><%= t2(j) %></TD>
</TR>
<TR>
<TD>売上高</TD>

```

```

<TD><%= sales %></TD>
</TR>
<TR>
<TD>支払経費合計</TD>
<TD><%= tcost %></TD>
</TR>
<TR>
<TD> 人件費等</TD>
<TD><%= Application("fixexpns") %></TD>
</TR>
<TR>
<TD> 増設陳列台費用</TD>
<TD><%= t3(j) %></TD>
</TR>
<TR>
<TD> 広告費</TD>
<TD><%= t4(j) %></TD>
</TR>
<TR>
<TD> 支払利息</TD>
<TD><%= t5(j) %></TD>
</TR>
<TR>
</TR>
<TR>
<TD>仕入枚数</TD>
<TD><%= t6(j) %></TD>
</TR>
<TR>
<TD>仕入金額</TD>
<TD><%= siire %></TD>
</TR>
<td>在庫枚数</td>
<td><%= t7(j) %></td>
</TR>
</TABLE><br><br>

```

```

<リスト10 inform2a>

```

```

# 12

```

```

<%

```

```

j = Application("joinno")

```



```
t1 = split(Application("n1"),",")
t2 = split(Application("n6"),",")
t3 = split(Application("n2"),",")
t4 = split(Application("n7"),",")
t5 = split(Application("n3"),",")
t6 = split(Application("n4"),",")
t7 = split(Application("n5"),",")
%>
```

<HR>

直前の値は次のようになっています。

<TABLE>

<TR>

<TD>現金</TD>

<TD><%= t1(j) %></TD>

<TD> | </TD>

<TD>買掛金</TD>

<TD><%= t2(j) %></TD>

</TR>

<TR>

<TD> (入金予定) </TD>

<TD> </TD>

<TD></TD>

</TR>

<TR>

<TD>売掛金</TD>

<TD><%= t3(j) %></TD>

</TR>

<TR>

<TD>受取手形 (今月) </TD>

<TD><%= t5(j) %></TD>

<TD></TD>

</TR>

<TR>

<TD>受取手形 (来月) </TD>

<TD><%= t6(j) %></TD>

<TD></TD>

</TR>

<TR>

<TD> </TD>

</TR>

<TR>

```

<TD>在庫数量</TD>
<TD><%= t7(j) %></TD>
<TD> </TD>
<TD>借入金残高</TD>
<TD><%= t4(j) %></TD>
</TR>
</TABLE><br><br>

```

13

[前サイクル最後の意思決定項目]

借入金増減額 (千円) : <INPUT TYPE = "text" NAME="debt" VALUE = "0">
(返済ならマイナスをつける)

[このサイクルの決定項目]

<TABLE BORDER="0">

<TR>

<TD>販売方法 : </TD>

</TR>

<TR>

<TD> 現金売</TD>

<TD><INPUT TYPE = "radio" NAME="sale" VALUE = "0.9"></TD>

<TD> 掛売</TD>

<TD><INPUT TYPE = "radio" NAME="sale" VALUE = "1" Checked></TD>

<TD> 手形売</TD>

<TD><INPUT TYPE = "radio" NAME="sale" VALUE = "1.1"></TD>

</TR>

<td>広告の有無 : </td>

</TR>

<TR>

<TD> 出す</TD>

<TD><INPUT TYPE = "radio" NAME="ad" VALUE = "1"></TD>

<TD> 出さない</TD>

<TD><INPUT TYPE = "radio" NAME="ad" VALUE = "0" Checked></TD>

</TR>

</TABLE>

<TABLE BORDER="0">

<TR>

<TD>陳列台台数 : </TD>

<TD><INPUT NAME="dai" TYPE="text" VALUE="4"></TD>

</TR>

<TR>

```
<TD>仕入枚数：</TD>
<TD><INPUT NAME="invol" TYPE="text" ></TD>
</TR>
</TABLE><br>
```

<リスト11 inform2b>

14

'すべての項目にデータが入力されているか？

If debt1 <> " " and sale1 <> " " and ad1 <> " " and dai1 <> " " and invol1 <> " " Then

Else

Application("errsw") = 1

Response.Write("入力漏れの項目があります。追加後、再送信してください

")

End If

15

'現金支払いと入金計算

in1 = split(Application("j1"),",")

in2 = split(Application("j2"),",")

in3 = split(Application("j3"),",")

in4 = split(Application("j4"),",")

in5 = split(Application("j5"),",")

in6 = split(Application("j6"),",")

in7 = split(Application("j7"),",")

in21 = split(Application("n1"),",")

in22 = split(Application("n2"),",")

in23 = split(Application("n3"),",")

in24 = split(Application("n4"),",")

in25 = split(Application("n5"),",")

in26 = split(Application("n6"),",")

in27 = split(Application("n7"),",")

'借入金の増減

debt2 = 0 + debt1

in27(jno1) = debt2 + in7(jno1)

in21(jno1) = debt2 + in1(jno1)

sa = 0 + in21(jno1) - in6(jno1)

If in21(jno1) < 0 Then

Response.Write("借入金の返済ができません。倒産です。

")

Application("errsw") = 1

wtr1(jno1) = "9"

'仕入代金の支払い

Elseif sa < 0 Then

```
Response.Write("仕入代金の支払いができません。倒産です。<br><br>")
Application("errsw") = 1
wtr1(jno1) = "9"
Else
    in21(jno1) = sa
    in26(jno1) = 0
End If
End If
If Application("errsw") = 0 Then
    ' 売上債権の回収
    in21(jno1) = 0 + in21(jno1) + in2(jno1) + in3(jno1)
    in23(jno1) = in4(jno1)
    in22(jno1) = 0
    in24(jno1) = 0
    Application("n1") = join(in21,",")
    Application("n2") = join(in22,",")
    Application("n3") = join(in23,",")
    Application("n4") = join(in24,",")
    Application("n7") = join(in27,",")
    ' 販売方法と広告の有無を記録
    in1 = split(Application("d1"),",")
    in2 = split(Application("d2"),",")
    in3 = split(Application("d3"),",")
    in4 = split(Application("d4"),",")
    in9 = split(Application("d5"),",")
    in1(jno1) = sale1
    in2(jno1) = ad1
    in3(jno1) = dai1
    in4(jno1) = invol1
    in9(jno1) = debt1
    Application("d1") = join(in1,",")
    Application("d2") = join(in2,",")
    Application("d3") = join(in3,",")
    Application("d4") = join(in4,",")
    Application("d5") = join(in9,",")
    ' 陳列台台数の修正
    dai1 = dai1 + 0
    If dai1 < 4 Then
        Response.Write("陳列台台数は4台を下回ることはできません。。<br><br>")
        dai1 = 4
    End If
End If
```

```

' 最大許容増加台数を設定
maxdai = 4
sa = + 4 + maxdai - dai1
' 指定台数は最大許容増加台数を超えているか？
' 指定台数と最大許容台数の小さい方（実際台数）を dai2 に記録
If sa < 0 Then
    Response.Write("陳列台台数を希望台数まで増加させることはできません。。 <br><br>")
    dai2 = maxdai + 4
Else
    dai2 = dai1 + 0
End If
' 先月から持ち越しの追加台数がある場合、今月の指定台数 dai2 は持ち越し台数を許容できるか？
in11 = split(Application("j11"),",")
sa = dai2 - 4 - in11(jno1)
If sa < 0 Then
    Response.Write("陳列台台数を減少させることはできません。。 <br><br>")
    dai2 = 4 + in11(jno1)
    sa = 0
End If
' sa > 0: 先月より台数が増加となる場合、または先月からの持ち越し台数がある場合、
' 将来の持ち越し必要台数を記録
If sa > 0 or in11(jno1) >0 Then
    in12 = split(Application("j12"),",")
    in13 = split(Application("j13"),",")
    in14 = split(Application("j14"),",")
    in15 = split(Application("j15"),",")
    in31 = split(Application("n11"),",")
    in32 = split(Application("n12"),",")
    in33 = split(Application("n13"),",")
    in34 = split(Application("n14"),",")
    in35 = split(Application("n15"),",")
    in31(jno1) = sa + in12(jno1)
    in32(jno1) = sa + in13(jno1)
    in33(jno1) = sa + in14(jno1)
    in34(jno1) = sa + in15(jno1)
    in35(jno1) = sa
    Application("n11") = join(in31,",")
    Application("n12") = join(in32,",")
    Application("n13") = join(in33,",")
    Application("n14") = join(in34,",")

```

```

Application("n15") = join(in35, ",")
End If
' 在庫可能枚数, 仕入可能枚数の計算
maxinvol = dai2 * 600 - IN5(jno1)
involl = 0 + involl
If involl > maxinvol Then
    Response.Write("希望枚数を仕入れることはできません。。 <br><br>")
    invol2 = maxinvol
Else
    invol2 = involl
End If
' 買掛金の計算
in26(jno1) = int(involl * 1.5)
Application("n6") = join(in26, ",")
' 在庫数量の更新
in25(jno1) = in5(jno1) + invol2
Application("n5") = join(in25, ",")
' 実際仕入数量, 陳列台数, 陳列台費用, 広告費, 支払利息の記録
in1 = split(Application("r1"), ",")
in2 = split(Application("r4"), ",")
in3 = split(Application("r5"), ",")
in4 = split(Application("r6"), ",")
in5 = split(Application("r7"), ",")
in1(jno1) = invol2
in2(jno1) = dai2
in3(jno1) = (dai2 - 4) * 10
in4(jno1) = 100 * ad1
in5(jno1) = int(in7(jno1) * 0.01)
Application("r1") = join(in1, ",")
Application("r4") = join(in2, ",")
Application("r5") = join(in3, ",")
Application("r6") = join(in4, ",")
Application("r7") = join(in5, ",")

```

リスト4 決算書の作成 Kessan.asp

```

<HTML>
<HEAD>
<META HTTP-EQUIV="Content-Type" CONTENT="text/html; charset=shift_jis">
<TITLE>決算の実施</TITLE>

```

```

<STYLE TYPE="text/css">
</STYLE>
</HEAD>
<BODY><br>
kessan.asp<br><br>
<% j = Application("joinno") %>
<%= "game-ID : " & Application("gameid") %>
<%= "サイクル番号 : " & Application("cycleno") %>
<%= "参加番号 : " & j & "<BR><BR><BR>" %>
<%
nn1 = split(Application("n1"),",")
nn2 = split(Application("n2"),",")
nn3 = split(Application("n3"),",")
nn4 = split(Application("n4"),",")
nn5 = split(Application("n5"),",")
nn6 = split(Application("n6"),",")
nn7 = split(Application("n7"),",")
cm1 = split(Application("cm1"),",")
cm2 = split(Application("cm2"),",")
cm3 = split(Application("cm3"),",")
bs1 = split(Application("bs1"),",")
bs3 = split(Application("bs3"),",")
bs5 = split(Application("bs5"),",")
bs6 = split(Application("bs6"),",")
wtr1 = split(Application("s1"),",")
' 税金支払い
cash1 = 0 + nn1(j) - bs3(j)
If cash1 < 0 Then
    Response.Write("税金支払いのための現金不足で倒産しました！<br><br>")
    wtr1(j) = "9"
    Application("s1") = join(wtr1,",")
End If
nn1(j) = cash1
Application("n1") = join(nn1,",")
' 決算用数値の計算
ukete = 0 + nn3(j) + nn4(j)
zaiko = int(nn5(j) * 1.5)
junsire = 0 + bs1(j) + cm2(j) - zaiko
sales = 0 + cm1(j)
expns = 0 + cm3(j)
profit1 = sales - junsire - expns - application("depress")

```

```

if profit1 < 0 Then
  zeil = 0
  junprfit = profit1
Else
  zeil = int(profit1/2)
  junprfit = profit1 - zeil
End if
If application("asetsw") = 0 Then
  fixasset1 = application("bs2") - application("depress")
  application("bs2") = fixasset1
  application("asetsw") = 1
Else
  fixasset1 = application("bs2")
End if
pastprofit = 0 + bs5(j) + bs6(j)
ltotal = cash1 + nn2(j) + ukete + zaiko + fixasset1
rtotal = zeil + nn6(j) + nn7(j) + application("bs4") + pastprofit + junprfit
%>

```

<HR>

貸借対照表

<TABLE BORDER="0">

<TR>

<TD>現金</TD>

<TD><%= cash1 %></TD>

<TD></TD>

<TD>買掛金</TD>

<TD><%= nn6(j) %></TD>

</TR>

<TR>

<TD>売掛金</TD>

<TD><%= nn2(j) %></TD>

<TD></TD>

<TD>借入金</TD>

<TD><%= nn7(j) %></TD>

</TR>

<TR>

<TD>受取手形</TD>

<TD><%= ukete %></TD>

<TD></TD>

<TD>納税引当金</TD>

ネットワーク・ゲーム構築モデルへのゲーム・コンテンツ組み込み事例（片山益男）

```

<TD><%= zeil %></TD>
</TR>
<TR>
<TD>商品</TD>
<TD><%= zaiko %></TD>
<TD></TD>
<TD>資本金</TD>
<TD><%= application("bs4") %></TD>
</TR>
<TR>
<TD>固定資産</TD>
<TD><%= fixasset1 %></TD>
<TD></TD>
<TD>繰越利益金</TD>
<TD><%= pastprofit %></TD>
</TR>
<TR>
<TD> </TD>
<TD></TD>
<TD></TD>
<TD>当期純利益</TD>
<TD><%= junprfit %></TD>
</TR>
<TR>
<TD>総資産</TD>
<TD><%= ltotal %></TD>
<TD></TD>
<TD>総資本</TD>
<TD><%= rtotal %></TD>
</TR>
</TABLE><br>
<HR>
損益計算書<br><br>
<TABLE BORDER="0">
<TR>
<TD>売上高</TD>
<TD><%= sales %></TD>
</TR>
<TR>
<TD>純仕入金額</TD>
<TD><%= junsiiire %></TD>

```

```

</TR>
<TR>
<TD>総支払費用</TD>
<TD><%= expns %></TD>
</TR>
<TR>
<TD>減価償却費</TD>
<TD><%= application("depress") %></TD>
</TR>
<TR>
<TD>税引前純利益</TD>
<TD><%= profit1 %></TD>
</TR>
</TABLE><br>

```

```
<HR>
```

```
<% 'DBへの記録
```

```

Set adoConn = Server.CreateObject("ADODB.Connection")
Set adoRecs= Server.CreateObject("ADODB.Recordset")
adoConn.Open "netgame"
adoRecs.Open "kessan", adoConn, 3, 3, 2
adoRecs.AddNew
adoRecs("gameid").Value = Application("gameid")
adoRecs("joinno").Value = j
adoRecs("cycleno").Value = Application("cycleno")
adoRecs("cash").Value = cash1
adoRecs("urikake").Value = nn2(j)
adoRecs("ukete").Value = ukete
adoRecs("zaiko").Value = zaiko
adoRecs("fixasst").Value = fixasset1
adoRecs("kaikake").Value = nn6(j)
adoRecs("kariire").Value = nn7(j)
adoRecs("zei").Value = zeil
adoRecs("sihon").Value = application("bs4")
adoRecs("kurikosiprfit").Value = pastprofit
adoRecs("profit").Value = junprfit
adoRecs("sales").Value = sales
adoRecs("netsiire").Value = junsiiire
adoRecs("totalexpend").Value = expns
adoRecs("depress").Value = application("depress")
adoRecs.Update
adoRecs.Close

```

```

adoConn.Close
' 決算用初期値の更新
cm1(j) = 0
cm2(j) = 0
cm3(j) = 0
bs1(j) = zaiko
bs3(j) = zeil
bs5(j) = pastprofit
bs6(j) = junprfit
Application("cm1") = join(cm1,",")
Application("cm2") = join(cm2,",")
Application("cm3") = join(cm3,",")
Application("bs1") = join(bs1,",")
Application("bs3") = join(bs3,",")
Application("bs5") = join(bs5,",")
Application("bs6") = join(bs6,",")
%>
</html>

```

リスト 5 モード変更までの時間問い合わせ time.asp

```

<HTML>
<HEAD>
<META HTTP-EQUIV="Content-Type" CONTENT="text/html; charset=shift_jis">
<TITLE>モード変更までの時間</TITLE>
<STYLE TYPE="text/css">
</STYLE>
</HEAD>
<BODY>
ゲームID:
<%= Application("gameid") %>
    サイクル番号：
<%= Application("cycleno") %><br><br>
<%= nmode = Application("pmode")
    If nmode = 0 Then
        Response.Write("ゲーム参加受付中です。<br><br>")
    ElseIf nmode = 1 Then
        Response.Write("入力モードです。<br><br>")
    ElseIf nmode = 2 Then
        Response.Write("問合せモードです。<br><br>")

```

```

Else
  Response.Write("ゲーム進行途中ではありません。<br><br>")
End If
%>
<%= "モード変更まで、あと " %>
<%= DateDiff("n",Now, Application("timelmt")) %>
分です (
<%= DateDiff("s",Now, Application("timelmt")) %>
秒) <br><br> (マイナスの値なら、すでにモードが変化しています。) <br>
</BODY>
</HTML>

```

リスト6 モード変更の締切時刻を現時点に変更 timechng.asp

```

<HTML>
<HEAD>
<META HTTP-EQUIV="Content-Type" CONTENT="text/html; charset=shift_jis">
<TITLE>締切時刻の変更</TITLE>
<STYLE TYPE="text/css">
</STYLE>
</HEAD>
<BODY>
ゲームID:
<%= Application("gameid") %>
  サイクル番号:
<%= Application("cycleno") %>
  モード番号:
<%= Application("pmode") %><br><br>
<% nmode = Application("pmode")
If nmode = 1 Then
  Application("nextlmt") = DateAdd("n", Application("seespan"), Now)
Else
  Application("nextlmt") = DateAdd("s", Application("inspan"), Now)
End If
Application("timelmt") = Now
%>
締切時刻を現時点 (
<%= Now %>
) に変更し、次の締切時刻を<br><br>
  <%= Application("nextlmt") %><br><br>

```

としました。

</BODY>

</HTML>